

## 3 (c) Complex geometries with McStas:

### A bi-spectral extraction guide system

E. Farhi <[farhi@ill.eu](mailto:farhi@ill.eu)> - Thursday Sept 19th 2013.

#### References:

Refer to the Workshops and Conferences McStas page at <<http://mcstas.org/workshop/>>

#### Scope:

In this exercise, we shall use two sources (cold and thermal), and see how to collect the flux from both. We start with simple components: a guide and an inclined mirror in front. Then we build a polygonal model of the guide with the inclined mirror inserted inside, and then add more than one of these inserted mirrors.

#### 3C.1: A bi-spectral source and a normal guide

1. In the Editor, select the menu item Edit/Insert/Instrument Template
2. Save your instrument as *Exercise3c\_bispectral.instr*
3. Add input parameters **DEFINE INSTRUMENT** *Exercise3c\_bispectral()*
4. Define two variables in the DECLARE block:  

```
double source_rand = 0;  
double source_colour = 0;
```
5. Use an **EXTEND** block after the **AT()** keyword of the Origin *Progress\_bar*, and assign *source\_rand = rand01(); source\_colour=0;* This casts a random number between 0 and 1 which will be used to select from which moderator should the neutron event be generated.
6. Insert a *Source\_gen* component instance 'thermal\_source' after the Origin, which emits neutrons between 0.1 and 10 Angs, 12x12 cm<sup>2</sup> area with a single Maxwellian 325K temperature and intensity 1e13 (leave T2,I2,T3,I3 as 0). The target window to send events to is 2 m away, 10x18 cm<sup>2</sup>. Locate it at the Origin.
7. Once inserted, add a **WHEN** condition to only activate the component for *source\_rand > 0.5*.
8. Then **EXTEND** the source to assign a colour for the generated neutrons with syntax  

```
if (SCATTERED) source_colour=1;
```
9. Add afterwards a **copy** of the first moderator, with T=50K, with condition *source\_rand <= 0.5* and locate at Y=+12cm above the first moderator. Rotate the 2nd moderator so that it point to the same target window (angle needed to move by 12 cm at 2 m distance).
10. Then **EXTEND** the 2nd source to assign a colour for the generated neutrons with syntax  



```
if (SCATTERED) source_colour=2;
```
11. Position a Monitor\_nD at 1.99 m from these sources, with 10x10 cm<sup>2</sup> area, that record the wavelength 'lambda' with automatic limits, as a function of the source\_colour. To do so, you may use

```

user1=source_colour, username1="Moderator index
[1=thermal, 2=cold]",
options="user1 limits=[0.9 2.1] bins=2, auto lambda"

```

12. Add a Guide component at 2 m from the Origin, width 10 cm, height 10 cm, length 10 m.
13. Add a copy of the previous Monitor\_nD instance at 10.01 m from the entrance of the guide.
14. You can additionally position a last wavelength monitor, such as Monitor\_nD(xwidth=.1, yheight=.1, options="auto lambda")
15. Save the simulation, and Run it.
16. Select the *Trace 3D view* with e.g. VRML or Matlab if you have installed InstantPlayer, FreeWRL or Matlab, or else default to PGPLOT.
17. Then re-run in Simulation mode and plot the two monitor histograms.
18. Is the cold source flux properly transmitted ?



### 3C.2: A mirror in front to reflect cold neutrons


We shall now insert an inclined mirror in front. The 10 m long guide will be slightly shifted away.

1. Move the insertion cursor just before the Guide instance, and insert a *Pol\_mirror* instance (in optics). The component is defined in the YZ plane, centred, so that we specify zwidth=2.88 m and yheight=0.10, and need to position and rotate it so that it comes right in front of the bottom guide side. If you can not manage to properly install this component in place, look at the solution at the end of this document.
2. Save the new instrument as *Exercise3c\_bispectral\_inclined.instr*
3. Run the simulation again and plot results.
4. Is there a change in the transmitted beam that reaches the end of the guide ?

### 3C.2: An inclined mirror inside a guide

For this step, we shall build a geometric representation of a focusing guide. To do so we can use advanced software such as MeshLab, AdMesh, FreeCAD. Such models, in OFF and PLY formats, can be used by the *Guide\_anyshape* component, as well as most sample components in McStas.

Such geometry files (STL and OBJ files are also following the same principles) first define some vertices (points in XYZ space). Each point is then assigned an index (usually starting from 0), and faces are lists of vertex indices, which form polygons.



We shall start with a rectangular shape, built 'by-hand'. Let's see how a cube is defined. We have 8 vertices, and 6 sides to define a guide. Now we should define the top face as inclined to direct to the cold moderator, whereas the lower face will be horizontal, directing to the thermal moderator.


The top face has then a vertical shift of  $12+12$  (2 moderators)- $10$ (guide) cm along  $2+2.88$  (empty+mirror), that is an angle of  $\text{atan}(.14/5.88)=1.36^\circ$ . Along the mirror length this results in a shift of about 6.8 cm, to add to the Y coordinates of points index 1 and 5, while the points index 2,4,6,7 have their Z at 2.88 m and the others have 0.1 (10 cm). In the end we have a focusing guide defined by the OFF :

```
OFF
# This is an Object File Format (geomview) to describe a 1 m^3 cube with sides only
# nb points, nb faces, void
8 4 0
# List 8 points coordinates
.1 0 0
0 .168 0
0 0 2.88
0 0 0
.1 0 2.88
.1 .168 0
.1 .1 2.88
0 .1 2.88
# List four side faces, all squared
4 0 5 6 4
4 1 5 6 7
4 3 0 4 2
4 3 1 7 2
```

which we save into an 'extractor.off' file (the faces first number 4 is the number of vertices per face).

Now we add 2 more points on the entry face of the extractor, with indices 8 and 9, and one new polygon which connects indices 2,4,9,8. The inclination of this polygon should have half the deviation angle  $1.36^\circ$  to act as a mirror, which represents a shift of  $\tan(1.36^\circ/2)*2.88=3.4$  cm.

```
OFF
# This is an Object File Format (geomview) to describe a 1 m^3 cube with sides only
# nb points, nb faces, void
10 5 0
# List 8 points coordinates
.1 0 0
0 .168 0
0 0 2.88
0 0 0
.1 0 2.88
```




```


.1 .168 0
.1 .1 2.88
0 .1 2.88
0 .034 0
.1 .034 0
# List four side faces, all squared
4 0 5 6 4
4 1 5 6 7
4 3 0 4 2
4 3 1 7 2
4 2 4 9 8

```

We can represent this shape using MeshLab or Geomview.



1. Once the geometry file has been assembled, you can use it with the Guide\_anyshape component. Replace the Pol\_mirror instance with a Guide\_anyshape using the 'extractor.off' file, with transmit option on, and m=5.
2. Save the instrument as *Exercise3c\_bispectral\_geometry.instr*
3. Run the instrument and generate a Trace 3D view (VRML is nice)
4. Run again in simulation mode.
5. Has the transmission for cold neutrons increased? What is the wavelength cut-off?



*Illustration 1: Result from the 3 simulations: green=guide far away ; red=inclined mirror (may be buggy); blue=inserted mirror in a focusing guide OFF*

---

```

DEFINE INSTRUMENT Exercise3c_BiSpectral_McStas()

DECLARE %{
  double source_rand = 0;
  double source_colour= 0;
%}

TRACE

COMPONENT Origin = Progress_bar()
  AT (0,0,0) ABSOLUTE
  EXTEND %{
    source_rand = rand01();
    source_colour= 0;
  %}

/* two moderators 12 cm apart. The cold one is rotated so that it illuminates the
same target window 10x10 cm2 at 2 m distance */
COMPONENT thermal_source = Source_gen(
  Lmin=.1, Lmax=10,
  xwidth=0.12, yheight=0.12,
  focus_xw=0.1, focus_yh=0.1, dist=2,
  T1=325, I1=1e13)
WHEN (source_rand > 0.5)
  AT(0,0,0) RELATIVE Origin
  EXTEND %{
    if (SCATTERED) source_colour=1;
  %}

COMPONENT cold_source = COPY(thermal_source) (
  T1=50, I1=1e13)
WHEN (source_rand <= 0.5)
  AT(0,0.12+tan(1.0*PI/180)*2.88,0) RELATIVE Origin
  ROTATED(atan2(.12,2)*180/PI, 0, 0) RELATIVE Origin
  EXTEND %{
    if (SCATTERED) source_colour=2;
  %}

/* put a Monitor at the guide entrance */
COMPONENT Mon_input = Monitor_nD(xwidth=0.1, yheight=0.18,
  user1=source_colour, username1="Moderator index [1=thermal, 2=cold]",
  options="user1 limits=[0.9 2.1] bins=2, auto lambda bins=20")
  AT(0,0,1.99) RELATIVE Origin

/* the Pol_mirror lies in the YZ plane, and is positioned at its centre */
/* so we translate by half dimensions, and rotate 90 deg around Z */
/* shift due to inclination is tan(1*PI/180)*2.88/2 */

COMPONENT inclined = Pol_mirror(
  zwidth = 2.88, yheight = 0.1,
  rUpPar={1.0, 0.0219, 3, 5.0, 0.003}, rDownPar={1.0, 0.0219, 3, 5.0, 0.003})
  AT(0,-0.1/2+tan(1*PI/180)*2.88/2,0.01+2.88/2) RELATIVE PREVIOUS
  ROTATED(1,0,90) RELATIVE PREVIOUS

COMPONENT guide = Guide(w1=0.1, h1=0.1, l=10)
  AT(0,0,0.01+2.88) RELATIVE PREVIOUS(2)

/* and a copy at the exit */
COMPONENT Mon_output = COPY(Mon_input)
  AT(0,0,10.01) RELATIVE PREVIOUS

END

```